

# Układy Cyfrowe i Systemy Wbudowane 1

## Język VHDL

dr inż. Jarosław Sugier  
Jaroslaw.Sugier@pwr.edu.pl  
K30W04ND03, pok. 227 C-3

IEEE 1076

1987, 93, 00, 01, 08, 14

- Specyfikacja  
- Symulacja (xas)  
- Synteza

*nie reasiniar  
wielkości  
liber*

*5. klasa*

*Architektura*

*1) Jednostki architektury  
opis układu = Entity + Architecture (s)*

*Hasła*

*- porty*

*- typ danych*

*- typ STD\_LOGIC '0', '1', 'X', ...*

*- w języku komponentów*

*- programie współzależne sygnałów*

*\* kolejność nie gra roli  
bo to parallel*

*uniwersalne*

```

library UNISIM;
use UNISIM.VComponents.all;
entity HalfAdder is
    port (
        A : in  STD_LOGIC;
        B : in  STD_LOGIC;
        S : out STD_LOGIC;
        C : out STD_LOGIC;
    );
end entity HalfAdder;

architecture Structural of HalfAdder is
begin
    XOR_gate : XOR2 port map ( A, B, S );
    AND_gate : AND2 port map ( A, B, C );
end architecture Structural;

architecture Dataflow of HalfAdder is
begin
    S <= A xor B;
    C <= A and B;
end architecture Dataflow;
    
```

## Literatura

- Język VHDL: M. Zwoliński(...) / K. Skahill(...) / IEEE Standard 1076 (PWR)
- Architektury układów PLD, CPLD: www...; [www.xilinx.com](http://www.xilinx.com)
- J. Kalisz „Podstawy elektroniki cyfrowej”, WKiŁ
- C. Zieliński „Podstawy projektowania układów cyfrowych”, PWN
- J. Pasierbiński, P. Zbysiński „Układy programowalne w praktyce”, WKiŁ
- T. Łuba (red.) „Synteza układów cyfrowych”, WKiŁ
- Pong P. Chu „RTL hardware design using VHDL”, J. Wiley

```


architecture Behavioral of HalfAdder is
begin
    process ( A, B )
    begin
        -- Sum
        if A /= B then
            S <= '1';
        else
            S <= '0';
        end if;
    end process;

    process ( A, B )
    begin
        -- Carry
        if A = '1' and B = '1' then
            C <= '1';
        else
            C <= '0';
        end if;
    end process;
end architecture Behavioral;
    
```

*... Mixed in one proces:  
process ( A, B )  
begin  
-- Sum  
if A /= B then  
S <= '1';  
else  
S <= '0';  
end if;  
-- Carry  
if A = '1' and B = '1' then  
C <= '1';  
else  
C <= '0';  
end if;  
end process;  
...*

## Porty i sygnały: tryby pracy portów


1) in - READ ONLY




Synteza:




2) out - WRITE ONLY



3) inout - bidirectional (3-state buffers, etc.)



4) buffer - „out with read capability”



## Porty i sygnały: sygnały wewnętrzne

```

entity AndNand is
    port (
        A : in  STD_LOGIC;
        B : in  STD_LOGIC;
        C : in  STD_LOGIC;
        WY_And : out STD_LOGIC;
        WY_Nand : out STD_LOGIC;
    );
end AndNand;

architecture DataflowBad of AndNand is
begin
    WY_And <= A and B and C;
    WY_Nand <= not WY_And;
end DataflowBad;
    
```

### Compilation:

HDLParas:1401 - Object WY\_And of mode OUT can not be read.

```

architecture DataflowOK of AndNand is
    signal Int_And : STD_LOGIC;
begin
    Int_And <= A and B and C;
    WY_And <= Int_And;
    WY_Nand <= not Int_And;
end DataflowOK;
    
```

## Wektory i napisy bitowe

Standardowy typ z biblioteki STD\_LOGIC\_1164:

```
type STD_LOGIC_VECTOR is array (NATURAL range <>)
of STD_LOGIC;
```

Użycie:

```
(...)
signal DataBus : STD_LOGIC_VECTOR( 7 downto 0 );
(...)

DataBus <= "10000000";
DataBus <= B"1000_0000";
DataBus <= X"80";
DataBus <= ( '1', '0', '0', '0', '0', '0', '0', '0' );
DataBus <= ( '1', others => '0' );
DataBus <= ( 7 => '1', others => '0' );

DataBus <= ( others => '0' );

HalfByte <= DataBus( 7 downto 4 );

MSB <= DataBus( 7 );
```

7

## Operator '&'

```
signal ASCII : STD_LOGIC_VECTOR( 7 downto 0 );
signal Digit : STD_LOGIC_VECTOR( 3 downto 0 );
(...)
ASCII <= "0011" & Digit; -- X"3" & Digit;
ASCII <= X"3" & Digit ( 3 ) & Digit ( 2 ) &
Digit ( 1 ) & Digit ( 0 );
(...)
-- These must be synchronous:
-- ...shift right
ASCII <= '0' & ASCII( 7 downto 1 );
-- ...arithmetic shift right:
ASCII <= ASCII( 7 ) & ASCII( 7 downto 1 );
-- ...rotate left:
ASCII <= ASCII( 6 downto 0 ) & ASCII( 7 );
```

8

## Klauzula generic

```
entity identifier is
generic ( parameter_declarations ); -- optional
port ( port_declarations ); -- optional
[ declarations ] -- optional
begin
[ statements ] -- optional
end entity identifier ;
```


Np.

```
entity Buf is
generic ( N : POSITIVE := 8; -- data width
Delay : DELAY_LENGTH := 2.5 ns );
port ( Input : in STD_LOGIC_VECTOR( N-1 downto 0 );
OE : in STD_LOGIC;
Output : out STD_LOGIC_VECTOR( N-1 downto 0 ) );
end entity Buf;
```

9

## Przypisanie sygnału

```
y <= x;
y <= (x1 and x2) or x3;
```



```
-- Domyślnie:
y <= x; <=> y <= inertial x after 0 ns;
y <= x after 4 ns; <=> y <= inertial x after 4 ns;

-- Przypisanie wielokrotne:
y <= '0', '1' after 100 ns, '0' after 120 ns;

-- Np. w opisach sekwencyjnych (powtarzanych w pętlach):
Clk <= '1', '0' after ClkPeriod / 2;
```

10

## Przypisanie warunkowe when...else

```
entity MUX_4 is
port( A, B, C, D : in STD_LOGIC;
Sel : in STD_LOGIC_VECTOR( 1 downto 0 );
Y : out STD_LOGIC );
end MUX_4;
architecture Dataflow of MUX_4 is
begin
Y <= A when Sel = "00" else
B when Sel = "01" else
C when Sel = "10" else
D;
D when Sel = "11" else
'X';
end Dataflow;
```

### • UWAGA!

```
Y <= '1' when A = '1' and B = '1' else
'0' when A = '0' and B = '0';
```

???

WARNING:Xst:737 - Found 1-bit latch for signal <Y>. Latches may be generated from incomplete case or if statements. We do not recommend the use of latches in FPGA/CPLD designs, as they may lead to timing problems.

11

## Przypisanie selektywne with ... select

```
(...)

architecture Dataflow2 of MUX_4 is
begin
with Sel select
Y <= A when "00",
B when "01",
C when "10",
D when "11",
'X' when others;
end Dataflow2;
```

• połączenie opcji (jako OR) - znakiem '|':

```
with Data( 2 downto 0 ) select
PE <= '1' when "001" | "010" | "100" | "111",
'0' when others;
```

12

## Instancje komponentów

JS UCiSW 1

```
entity XOR_2WE is
    generic( Tp : DELAY_LENGTH );
    port ( I1, I2 : in STD_LOGIC;
           O : out STD_LOGIC);
end XOR_2WE;
architecture A of XOR_2WE is
begin
    O <= I1 xor I2 after Tp;
end A;


entity AND_2WE is
    generic( Tp : DELAY_LENGTH );
    port ( I1, I2 : in STD_LOGIC;
           O : out STD_LOGIC);
end AND_2WE;
architecture A of AND_2WE is
begin
    O <= I1 and I2 after Tp;
end A;

entity HalfAdder is
    (...)
architecture Structural of HalfAdder is
    component XOR_2WE is
        generic( Tp : DELAY_LENGTH );
        port ( I1, I2 : in STD_LOGIC; O : out STD_LOGIC);
    end component;
    component AND_2WE is
        generic( Tp : DELAY_LENGTH );
        port ( I1, I2 : in STD_LOGIC; O : out STD_LOGIC);
    end component;
begin
    XOR_gate : XOR_2WE generic map( 5 ns ) port map( A, B, S );
    AND_gate : AND_2WE generic map( Tp => 3 ns )
        port map( O=>C, I1=>A, I2=>B );
end architecture Structural;
```

13

Niepodłączone WE (open) a symulacja behawioralna:

JS UCiSW 1



14

## Przykład 1

ISE: plik z pobudzeniami testowymi

JS UCiSW 1

```
1 LIBRARY ieee;
2 USE ieee.std_logic_1164.ALL;
3 USE ieee.numeric_std.ALL;
4 LIBRARY UNISIM;
5 USE UNISIM.vcomponents.ALL;
6
7 ENTITY Top_sch_tbw IS
8 END Top_sch_tbw;
9
10 ARCHITECTURE behavioral OF Top_sch_tbw IS
11
12     COMPONENT Top
13     PORT ( We2 : IN STD_LOGIC;
14           We1 : IN STD_LOGIC;
15           Wy : OUT STD_LOGIC);
16     END COMPONENT;
17
18     SIGNAL We2 : STD_LOGIC;
19     SIGNAL We1 : STD_LOGIC;
20     SIGNAL Wy : STD_LOGIC;
21
22 BEGIN
23
24     UUT: Top PORT MAP(
25         We2 => We2,
26         We1 => We1,
27         Wy => Wy
28     );
29
30     WE1 <= '0', '1' after 100 ns, '0' after 300 ns;
31     WE2 <= '0', '1' after 200 ns, '0' after 400 ns;
32
33
34 END;
```

15

## Przykład 2

ISE: plik vhf na podstawie sch

JS UCiSW 1

```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;
library UNISIM;
use UNISIM.vcomponents.all;

entity SimpleSch is
    port ( Clk : in STD_LOGIC;
           E1 : in STD_LOGIC;
           E2 : in STD_LOGIC;
           Reset : in STD_LOGIC;
           Y : out STD_LOGIC);
end SimpleSch;

architecture BEHAVIORAL of SimpleSch is
    signal XLXN_1 : STD_LOGIC;
    signal XLXN_2 : STD_LOGIC;
    signal Y_DUMMY : STD_LOGIC;

    component FDR
        port ( C : in STD_LOGIC;
              D : in STD_LOGIC;
              R : in STD_LOGIC;
              Q : out STD_LOGIC);
    end component;
    component XOR2
        port ( I0 : in STD_LOGIC;
              I1 : in STD_LOGIC;
              O : out STD_LOGIC);
    end component;
    component AND2
        port ( I0 : in STD_LOGIC;
              I1 : in STD_LOGIC;
              O : out STD_LOGIC);
    end component;

    Y <= Y_DUMMY;
    XLXI_1 : FDR
        port map (C=>Clk,
                  D=>XLXN_1,
                  R=>Reset,
                  Q=>Y_DUMMY);
    XLXI_2 : XOR2
        port map (I0=>XLXN_2,
                  I1=>Y_DUMMY,
                  O=>XLXN_1);
    XLXI_3 : AND2
        port map (I0=>E2,
                  I1=>E1,
                  O=>XLXN_2);

begin
    XLXN_1 <= (E1 and E2) xor Y_DUMMY;
    XLXN_2 <= (E1 and E2) and Y_DUMMY;
```

16

## Instrukcja generacji

JS UCiSW 1

### a) for ... generate

*Np.*

```
entity FullAdder is
    port ( A, B : in STD_LOGIC_VECTOR( 7 downto 0 );
           CI : in STD_LOGIC;
           S : out STD_LOGIC_VECTOR( 7 downto 0 );
           CO : out STD_LOGIC);
end FullAdder;

architecture Dataflow of FullAdder is
    signal Cint : STD_LOGIC_VECTOR( 8 downto 0 );
begin
    lb: for i in 0 to 7 generate
        S(i) <= A(i) xor B(i) xor Cint(i);
        Cint(i + 1) <= ( A(i) and B(i) ) or
            ( A(i) and Cint(i) ) or ( B(i) and Cint(i) );
    end generate;
    Cint( 0 ) <= CI;
    CO <= Cint( 8 );
end Dataflow;
```

17

### b) if ... generate

```
label: if condition generate -- label required
    block_declarative_items \ optional
begin
    concurrent_statements
end generate label;
```

18

JS UCiSW 1

### Instrukcja procesu

```
[label:] process [ ( sensitivity_list ) ] [ is ]
    [ declarative_items ]
begin
    sequential_statements
end process [ label ];
```

Np. było:

```
process( A, B ) is
begin
    if A /= B then
        S <= '1';
    else
        S <= '0';
    end if;
    if A = '1' and B = '1' then
        C <= '1';
    else
        C <= '0';
    end if;
end process;
```

19

JS UCiSW 1

### Instrukcje sekwencyjne

( ... )

Instrukcja wait:

```
wait for 10 ns;           -- timeout
wait until clk = '1';     -- warunek logiczny
wait until A > B and ( S1 or S2 );
wait on sig1, sig2;       -- lista wrażliwości
```

wait until ... – czeka na zdarzenie, tj. zmianę sygnału (zawsze wstrzyma wykonanie procesu!); jeśli nie o to chodzi to trzeba np. dodać warunek:

```
if Busy /= '0' then
    wait until Busy = '0';
end if;
```

20

JS UCiSW 1

```
[ label: ] if condition1 then
    statements
elseif condition2 then \_ optional
    statements
...
else \_ optional
    statements
end if [ label ] ;
```

```
architecture DF of MUX_4 is
begin
    Y <= A when Sel = "00" else
        B when Sel = "01" else
        C when Sel = "10" else
        D when Sel = "11" else
        'X';
end DF;
```

```
architecture DF_Eq of MUX_4 is
begin
    process ( Sel, A, B, C, D )
    begin
        if Sel = "00" then
            Y <= A;
        elsif Sel = "01" then
            Y <= B;
        elsif Sel = "10" then
            Y <= C;
        elsif Sel = "11" then
            Y <= D;
        else
            Y <= 'X';
        end if;
    end process;
end DF_Eq;
```

21

JS UCiSW 1

```
[ label: ] case expression is
    when choice1 =>
        statements
    when choice2 => \_ opt.
        statements
    ...
    when others => \_ opt. if all choices
        statements / covered
end case [ label ] ;
```

```
architecture DF2 of MUX is
begin
    with Sel select
        Y <= A when "00",
            B when "01",
            C when "10",
            D when "11",
            'X' when others;
end DF2;
```

```
architecture DF2_Eq of MUX_4 is
begin
    process ( Sel, A, B, C, D )
    begin
        case Sel is
            when "00" => Y <= A;
            when "01" => Y <= B;
            when "10" => Y <= C;
            when "11" => Y <= D;
            when others => Y <= 'X';
        end case;
    end process;
end DF2_Eq;
```

22

JS UCiSW 1

```
[ label: ] loop
    statements -- use exit to abort
end loop [ label ] ;

[ label: ] for variable in range loop
    statements
end loop [ label ] ;

[ label: ] while condition loop
    statements
end loop [ label ] ;
```

```
next;
next outer_loop; -- label of loop instr.
next when A > B;
next this_loop when C = D or A > B;
```

```
exit;
exit outer_loop;
exit when A > B;
exit this_loop when C = D or A > B;
```

23

JS UCiSW 1

### Inne instrukcje

**Instrukcja null**

Np.:

```
case (...) is
    (...)
    when others => null;
end case;
```

**Instrukcja report**

```
report "Message" [ severity SevLevel ];
type SEVERITY_LEVEL is (NOTE, WARNING, ERROR, FAILURE);
Np.
    report "Parity bit error" severity WARNING;
```

**Instrukcja assert**

```
assert boolean_cond [report "Msg"] [severity SevLevel];
⇒ Domyślnie: SevLevel = ERROR
```

**Instrukcja return**

Konieczna w funkcjach, opcjonalna w procedurach.

**Wywołanie procedury**

24

## Sygnaly synchroniczne

JS UCiSW 1

```
entity DFF is
  port ( D : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC );
end DFF;

architecture RTL of DFF is
begin
  process ( Clk )
  begin
    if Clk'Event and Clk = '1' then
      Q <= D;
    end if;
  end process;
end architecture;
```

```
entity TFF is
  port ( T : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC );
end TFF;

architecture RTL of TFF is
  signal Q_int : STD_LOGIC := '0';
begin
  Q <= Q_int;
  process ( Clk )
  begin
    if Clk'Event and Clk = '1' then
      if T = '1' then
        Q_int <= not Q_int;
      end if;
    end if;
  end process;
end architecture;
```

• Pakiet STD\_LOGIC\_1164:

```
function rising_edge (signal s : STD_ULOGIC) return BOOLEAN;
function falling_edge (signal s : STD_ULOGIC) return BOOLEAN;

if Clk'Event and Clk = '1' then... => if rising_edge(Clk) then...
```

25

## Clock Enable (~FDE):

JS UCiSW 1

```
entity DFF_E is
  port( D : in STD_LOGIC;
        CE : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC );
end DFF_E;

architecture RTL of DFF_E is
begin
  process ( Clk )
  begin
    if rising_edge( Clk ) then
      if CE = '1' then
        Q <= D;
      end if;
    end if;
  end process;
end architecture;
```

26

## Asynchronous Clear + Enable: (FDCE) Synchronous Reset + Enable: (FDRE)

JS UCiSW 1

```
entity DFF_CE is
  port( D : in STD_LOGIC;
        Clr : in STD_LOGIC;
        CE : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC );
end DFF_CE;

architecture RTL of DFF_CE is
begin
  process ( Clk, Clr )
  begin
    if Clr = '1' then
      Q <= '0';
    elsif rising_edge(Clk) then
      if CE = '1' then
        Q <= D;
      end if;
    end if;
  end process;
end architecture;
```

```
entity DFF_RE is
  port( D : in STD_LOGIC;
        Rst : in STD_LOGIC;
        CE : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC );
end DFF_RE;

architecture RTL of DFF_RE is
begin
  process ( Clk )
  begin
    if rising_edge( Clk ) then
      if Rst = '1' then
        Q <= '0';
      elsif CE = '1' then
        Q <= D;
      end if;
    end if;
  end process;
end architecture;
```

27

## ERROR:Xst:827: Signal Q cannot be synthesized, bad synchronous description (...)

JS UCiSW 1

FDCE OK

ERROR: bad synchronous descr! (FDEC?)

ERROR: bad synchronous descr! (FDEC?)

⇒ Symulacja behawioralna nie widzi problemu

FDRE OK

Synthesizable to ~FDER, not recommended

Synthesizable to FDRE, bad style!

28

## Rejestr przesuwany:

JS UCiSW 1

```
entity SReg8b is
  port ( Din : in STD_LOGIC;
        Clk : in STD_LOGIC;
        Q : out STD_LOGIC_VECTOR( 7 downto 0 ) );
end SReg8b;

architecture RTL of SReg8b is
  signal iQ : STD_LOGIC_VECTOR( 7 downto 0 );
begin
  Q <= iQ;
  process ( Clk )
  begin
    if rising_edge( Clk ) then
      iQ( 7 downto 0 ) <= iQ( 6 downto 0 ) & Din;
    end if;
  end process;
end architecture;
```

29

## Np.

JS UCiSW 1

architecture RTL of SimpleSch is

signal D, Q : STD\_LOGIC;

begin

D <= ( E1 and E2 ) xor Q;

process( Clk )

begin

if rising\_edge( Clk ) then

if Reset = '1' then

Q <= '0';

else

Q <= D;

end if;

end if;

end process;

Y <= Q;

end RTL;

30

### Licznik binarny z asynchronicznym kasowaniem:

JS UCiSW 1

```
library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity counters_1 is
    port(C, CLR : in STD_LOGIC;
          Q : out STD_LOGIC_VECTOR(3 downto 0));
end counters_1;

architecture archi of counters_1 is
    signal tmp: UNSIGNED(3 downto 0);
begin
    process ( C, CLR )
    begin
        if CLR = '1' then
            tmp <= "0000";
        elsif rising_edge( C ) then
            tmp <= tmp + 1;
        end if;
    end process;
    Q <= STD_LOGIC_VECTOR( tmp );
end archi;
```

31

### Licznik modulo (z zerowaniem i sygnałem zezwalającym):

JS UCiSW 1

```
(...)
process ( Clk )
begin
    if rising_edge( Clk ) then
        if Rst = '1' then
            tmp <= "0000";
        elsif CE = '1' then
            if tmp = "1001" then
                tmp <= "0000";
            else
                tmp <= tmp + 1;
            end if;
        end if;
    end if;
end process;
(...)
```

32

### Licznik ładowalny (asynchronicznie):

JS UCiSW 1

```
process ( Clk, ALOAD, D )
begin
    if ALOAD = '1' then
        tmp <= D;
    elsif rising_edge( Clk ) then
        tmp <= tmp + 1;
    end if;
end process;
```

### Licznik rewersyjny:

```
process ( Clk, CLR )
begin
    if CLR = '1' then
        tmp <= "0000";
    elsif rising_edge( Clk ) then
        if UP_DOWN = '1' then
            tmp <= tmp + 1;
        else
            tmp <= tmp - 1;
        end if;
    end if;
end process;
```

33

### Zatrask:

JS UCiSW 1

```
entity latches_1 is
    port(G, D : in std_logic;
          Q : out std_logic);
end latches_1;

architecture archi of latches_1 is
begin
    process (G, D)
    begin
        if (G='1') then
            Q <= D;
        end if;
    end process;
end archi;
```

34

### Zatrask z asynchronicznym kasowaniem:

JS UCiSW 1

```
architecture archi of latches_2 is
begin
    process (CLR, D, G)
    begin
        if (CLR='1') then
            Q <= '0';
        elsif (G='1') then
            Q <= D;
        end if;
    end process;
end archi;
```

```
Q <= '0' when CLR = '1' else
    D when G = '1';
```


35

### Bufor trójstanowy:

JS UCiSW 1


```
entity three_st_2 is
    port(T : in std_logic;
          I : in std_logic;
          O : out std_logic);
end three_st_2;

architecture archi of three_st_2 is
begin
    O <= I when (T='0') else 'Z';
end archi;
```




36

## Maszyny stanów




X9067

37



38


## FSM in XST (Xilinx Synthesis Tool)



39

## Opisy dla symulacji


Przykład 1: Nadawanie bajtu do odbiornika RS-232 (UUT: port in RS\_RX)



40


## Opisy dla symulacji

Przykład 1: Nadawanie bajtu do portu RS-232 (UUT: port in RS\_RX)



41

Przykład 2: Odbiór bajtu z nadajnika RS-232 (UUT: port out RS\_TX)



42

Przykład 3: Nadawanie bajtów do portu PS/2 (UUT: PS2\_Data, PS2\_Clk) JS UCiSW 1

```

(...)
process
begin
  procedure TransmPS2( Byte : STD_LOGIC_VECTOR( 7 downto 0 ) ) is
    variable Frame : STD_LOGIC_VECTOR( 10 downto 0 ) := "11" & Byte & '0';
  begin
    -- Parity calculation
    for i in 0 to 7 loop
      Frame( 9 ) := Frame( 9 ) xor Byte( i );
    end loop;
    -- Transmission of the frame; Freq.Clk = 10kHz (Tclk = 100 us)
    for i in 0 to 10 loop
      PS2_Data <= Frame( i );
      wait for 5 us;
      PS2_Clk <= '0', '1' after 50 us;
      wait for 95 us; -- 100us per loop
    end loop;
  end procedure;
begin
  PS2_Data <= '1';
  PS2_Clk <= '1';
  wait for 15 us;
  TransmPS2( X"F0" );
  wait for 200 us;
  TransmPS2( X"81" );
  wait; -- will wait forever
end process;
(...)

```

43

Pakiet STANDARD JS UCiSW 1

```

type INTEGER is range --usually typical INTEGER-- ;
subtype NATURAL is INTEGER range 0 to INTEGER'HIGH;
subtype POSITIVE is INTEGER range 1 to INTEGER'HIGH;
type REAL is range --usually double precision f.p.-- ;
type BOOLEAN is (FALSE, TRUE);
type CHARACTER is ( --256 characters-- );
type STRING is array (POSITIVE range <>) of CHARACTER;
type BIT is ('0', '1');
type TIME is range --implementation defined-- ;

units
  fs;          -- femtosecond
  ps = 1000 fs; -- picosecond
  ns = 1000 ps; -- nanosecond
  us = 1000 ns; -- microsecond
  ms = 1000 us; -- millisecond
  sec = 1000 ms; -- second
  min = 60 sec; -- minute
  hr = 60 min; -- hour
end units;
subtype DELAY_LENGTH is TIME range 0 fs to TIME'HIGH;

```

44

Operatory JS UCiSW 1

<b>**</b>	exponentiation,	numeric <b>**</b> integer,	result numeric
<b>abs</b>	absolute value,	<b>abs</b> numeric,	result numeric
<b>not</b>	complement,	<b>not</b> logic or boolean,	result same
<b>*</b>	multiplication,	numeric <b>*</b> numeric,	result numeric
<b>/</b>	division,	numeric <b>/</b> numeric,	result numeric
<b>mod</b>	modulo,	integer <b>mod</b> integer,	result integer
<b>rem</b>	remainder,	integer <b>rem</b> integer,	result integer
<b>+</b>	unary plus,	<b>+</b> numeric,	result numeric
<b>-</b>	unary minus,	<b>-</b> numeric,	result numeric
<b>+</b>	addition,	numeric <b>+</b> numeric,	result numeric
<b>-</b>	subtraction,	numeric <b>-</b> numeric,	result numeric
<b>&amp;</b>	concatenation,	array or element,	result array

45

JS UCiSW 1

<b>sll</b>	shift left logical,	log. array <b>sll</b> integer,	result same
<b>srl</b>	shift right log.,	log. array <b>srl</b> integer,	result same
<b>sla</b>	shift left arith.,	log. array <b>sla</b> integer,	result same
<b>sra</b>	shift right arith.,	log. array <b>sra</b> integer,	result same
<b>rol</b>	rotate left,	log. array <b>rol</b> integer,	result same
<b>ror</b>	rotate right,	log. array <b>ror</b> integer,	result same
<b>=</b>	equality,		result boolean
<b>/=</b>	inequality,		result boolean
<b>&lt;</b>	less than,		result boolean
<b>&lt;=</b>	less than or equal,		result boolean
<b>&gt;</b>	greater than,		result boolean
<b>&gt;=</b>	greater than or equal,		result boolean
<b>and</b>	logical and,	log. array or boolean,	result same
<b>or</b>	logical or,	log. array or boolean,	result same
<b>nand</b>	logical nand,	log. array or boolean,	result same
<b>nor</b>	logical nor,	log. array or boolean,	result same
<b>xor</b>	logical xor,	log. array or boolean,	result same
<b>xnor</b>	logical xnor,	log. array or boolean,	result same

46

Pakiet STD\_LOGIC\_1164 JS UCiSW 1

```

library IEEE;
use IEEE.STD_LOGIC_1164.all;

type STD_ULOGIC is ( 'U', -- Uninitialized
                    'X', -- Forcing Unknown
                    '0', -- Forcing 0
                    '1', -- Forcing 1
                    'Z', -- High Impedance
                    'W', -- Weak Unknown
                    'L', -- Weak 0
                    'H', -- Weak 1
                    '-' -- Don't care
                    );

```

```

type STD_ULOGIC_VECTOR is array ( NATURAL range <> ) of STD_ULOGIC;

```

47

JS UCiSW 1

```

function resolved ( s : STD_ULOGIC_VECTOR ) return STD_ULOGIC;

constant resolution_table : stdlogic_table := (
--
-- | U  X  0  1  Z  W  L  H  -  | |
--
( 'U', 'U', 'U', 'U', 'U', 'U', 'U', 'U', 'U' ), -- | U |
( 'U', 'X', 'X', 'X', 'X', 'X', 'X', 'X', 'X' ), -- | X |
( 'U', '0', '0', '0', '0', '0', '0', '0', '0' ), -- | 0 |
( 'U', '1', '1', '1', '1', '1', '1', '1', '1' ), -- | 1 |
( 'U', 'Z', 'Z', 'Z', 'Z', 'Z', 'Z', 'Z', 'Z' ), -- | Z |
( 'U', 'W', 'W', 'W', 'W', 'W', 'W', 'W', 'W' ), -- | W |
( 'U', 'L', 'L', 'L', 'L', 'L', 'L', 'L', 'L' ), -- | L |
( 'U', 'H', 'H', 'H', 'H', 'H', 'H', 'H', 'H' ), -- | H |
( 'U', 'X', 'X', 'X', 'X', 'X', 'X', 'X', 'X' ) -- | - |
);
(...)

```

```

-- *** industry standard logic type ***
--
subtype STD_LOGIC is resolved STD_ULOGIC;
type STD_LOGIC_VECTOR is array ( NATURAL range <> ) of STD_LOGIC;

```

48



```

-- truth table for "and" function
constant and_table : stdlogic_table := (
--
| U X 0 1 Z W L H - |
( 'U', 'U', '0', '0', '0', '0', '0', '0', '0', '0' ), -- | U |
( 'U', 'X', '0', '0', '0', '0', '0', '0', '0', '0' ), -- | X |
( '0', '0', '0', '0', '0', '0', '0', '0', '0', '0' ), -- | 0 |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | 1 |
( 'U', 'X', '0', '0', '1', '0', '0', '0', '0', '0' ), -- | Z |
( 'U', 'X', '0', '0', '0', '1', '0', '0', '0', '0' ), -- | W |
( 'U', 'X', '0', '0', '0', '0', '1', '0', '0', '0' ), -- | L |
( 'U', 'X', '0', '0', '0', '0', '0', '1', '0', '0' ), -- | H |
( 'U', 'X', '0', '0', '0', '0', '0', '0', '1', '0' ), -- | - |
);
-- truth table for "or" function
constant or_table : stdlogic_table := (
--
| U X 0 1 Z W L H - |
( 'U', 'U', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | U |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | X |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | 0 |
( '1', '1', '1', '1', '1', '1', '1', '1', '1', '1' ), -- | 1 |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | Z |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | W |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | L |
( '1', '1', '1', '1', '1', '1', '1', '1', '1', '1' ), -- | H |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | - |
);

```

49

```

-- truth table for "xor" function
constant xor_table : stdlogic_table := (
--
| U X 0 1 Z W L H - |
( 'U', 'U', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | U |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | X |
( 'U', 'X', '0', '1', '0', '0', '0', '0', '0', '0' ), -- | 0 |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | 1 |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | Z |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | W |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | L |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | H |
( 'U', 'X', '1', '1', '0', '0', '0', '0', '0', '0' ), -- | - |
);
-- truth table for "not" function
constant not_table : stdlogic_table := (
--
| U X 0 1 Z W L H - |
( 'U', 'X', '1', '0', '0', '0', '0', '0', '0', '0' );

```

50

## Pakiet NUMERIC\_STD (IEEE Std. 1076.3)

- Pre 1076.3: Synopsys libraries std\_logic\_unsigned/ std\_logic\_signed – now obsolete, but used to be *defacto* industry standard

```

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;
...
// The two NUMERIC_STD types:
signal A_u : UNSIGNED(3 downto 0); -- "s.l.v." interpreted as unsigned
signal B_s : SIGNED(3 downto 0); -- "s.l.v." interpreted as signed
...
A_u <= "1111"; -- 15 decimal
B_s <= "1111"; -- -1 decimal
...
A_u <= A_u + 1; -- overloaded '+' operator; also '-', '<'...

```

51

```

-- Obligatory explicit type conversions:
std_l_vec <= STD_LOGIC_VECTOR( unsigned | signed );
unsigned <= UNSIGNED( std_l_vec );
signed <= SIGNED( std_l_vec );

-- Conversion from/to integer types:
unsigned <= TO_UNSIGNED( 128, 8 ); -- value, vector_size
signed <= TO_SIGNED( -7, 8 );
std_l_vec <= STD_LOGIC_VECTOR( TO_UNSIGNED( 1076, 32 ) );
int <= TO_INTEGER( unsigned | signed );
int <= TO_INTEGER( (UN)SIGNED( std_l_vec ) );

-- Carry out in additions
Result_8b <= Arg1_8b + Arg2_8b; -- carry out is lost
Result_9b <= ('0' & Arg1_8b) + ('0' & Arg2_8b);
Carry_Out <= Result_9b( 8 ); -- OK
-- Note: Result_9b <= Arg1_8b + Arg2_8b will not work!

```

52

## Atrybuty

NamedEntity'AttrName[ (ParameterList) ]

### Atrybuty wektorów / typów wektorowych

A'LEFT is the leftmost idx of array A or constrained array type.

A'RIGHT is the rightmost idx of array A or constrained array type.

```

type bit_array is array (-15 to 15) of bit;
variable L: INTEGER := bit_array'Left; -- L has a value of -15
...
for i in some_vector'left to some_vector'left + 4 loop
... browse the first 5 elements ...

```

A'HIGH is the highest idx of array A or constrained array type.

A'LOW is the lowest idx of array A or constrained array type.

```
variable H: INTEGER := bit_array'High; -- H has a value of 15
```

A'LENGTH is the integer value of the number of elements in array A.

```
variable LEN: INTEGER := bit_array'Length -- LEN has a value of 31
```

A'RANGE is the range A'LEFT to A'RIGHT or A'LEFT downto A'RIGHT.

```
for i in some_vector'Range loop
... browse all elements ...
```

53

A'REVERSE\_RANGE is the range of A with to and downto reversed.

A'ASCENDING is boolean true if range of A defined with to.

### Atrybuty typów

T'LEFT is the leftmost value of type T (largest if downto)

T'RIGHT is the rightmost value of type T (smallest if downto)

```
variable V: INTEGER := INTEGER'Left; -- value of -2,147,483,648
```

T'HIGH is the highest value of type T.

T'LOW is the lowest value of type T.

```
variable V: REAL := REAL'High; -- 1.701411e+308
```

T'POS(X) is (zero-based) integer position of X in the discrete type T.

```
type state_type is (Init, Hold, Strobe, Read, Idle);
variable P: integer := state_type'Pos( Hold ); -- value of 1
...
-- State number to the out port (synthesizable by XST!)
DbgOutput <= to_unsigned( state_type'Pos( state ), 3 );
```

T'VAL(X) is the value of discrete type T at integer position X.

```
variable V: state_type := state_type'Val( 2 ); -- value of Strobe
```

54

	JS UCiSW 1
T'PRED(X) is the value of discrete type T that is the predecessor of X.	
T'SUCC(X) is the value of discrete type T that is the successor of X.	
<b>variable</b> V: state_type := state_type'Succ(Init); -- value of Hold	
T'LEFTOF(X) is the value of discrete type T that is left of X.	
T'RIGHTOF(X) is the value of discrete type T that is right of X.	
-- Different from 'Pred / 'Succ only in a subtype	
-- which changed order of the base type	
T'ASCENDING is boolean true if range of T defined with to.	
T'IMAGE(X) is a string representation of X that is of type T.	
<b>report</b> INTEGER'Image( to_integer( unsigned( slv ) ) );	
T'VALUE(X) is a value of type T converted from the string X.	
<b>constant</b> P1: REAL := REAL'Value( "3.141" );	
T'BASE is the base type of type T	
POSITIVE'Base'Left -- INTEGER'Left or -2,147,483,648	
	55

	JS UCiSW 1
<b>Np. 1) Konwersja STRING na literal bitowy</b>	
<b>constant</b> slvHello : STD_LOGIC_VECTOR(39 downto 0)	
:= X"48_65_6C_6C_6F"; -- "Hello"	
Albo:	
<b>function</b> string_to_slv( S : STRING ) <b>return</b> STD_LOGIC_VECTOR <b>is</b>	
<b>constant</b> strLength : NATURAL := S'Length;	
<b>constant</b> Snorm : STRING( 1 to strLength ) := S;	
<b>variable</b> Result: STD_LOGIC_VECTOR(strLength * 8 - 1 downto 0);	
<b>begin</b>	
<b>for</b> i <b>in</b> 0 <b>to</b> strLength - 1 <b>loop</b>	
Result(i * 8 + 7 <b>downto</b> i * 8) :=	
STD_LOGIC_VECTOR( to_unsigned(	
CHARACTER'Pos( Snorm(strLength - i) ), 8) );	
<b>end loop;</b>	
<b>return</b> Result;	
<b>end function;</b>	
(...)	
<b>constant</b> slvHello : STD_LOGIC_VECTOR :=	
string_to_slv( "Hello" );	
	56

	JS UCiSW 1
<b>Np. 2) Konwersja slv Hex na slv ASCII</b>	
X"0123A" → X"30_31_32_33_41"	
<b>function</b> slvHex_to_slvASCII( slvHex : STD_LOGIC_VECTOR )	
<b>return</b> STD_LOGIC_VECTOR <b>is</b>	
<b>constant</b> NoDigits : INTEGER := slvHex'Length / 4;	
<b>variable</b> Hex : STD_LOGIC_VECTOR( 3 <b>downto</b> 0);	
<b>variable</b> ASCII : STD_LOGIC_VECTOR( 7 <b>downto</b> 0);	
<b>variable</b> slvASCII : STD_LOGIC_VECTOR(NoDigits * 8 - 1 <b>downto</b> 0);	
<b>begin</b>	
<b>for</b> i <b>in</b> 0 <b>to</b> NoDigits - 1 <b>loop</b>	
Hex := slvHex(i * 4 + 3 <b>downto</b> i * 4);	
<b>case</b> Hex <b>is</b>	
<b>when</b> X"A" => ASCII := X"41";	
-- STD_LOGIC_VECTOR( to_unsigned( Character'POS('A') ) );	
<b>when</b> X"B" => ASCII := X"42";	
<b>when</b> X"C" => ASCII := X"43";	
<b>when</b> X"D" => ASCII := X"44";	
<b>when</b> X"E" => ASCII := X"45";	
<b>when</b> X"F" => ASCII := X"46";	
<b>when others</b> => ASCII := X"3" & Hex; -- '0'...'9'	
<b>end case;</b>	
slvASCII(i * 8 + 7 <b>downto</b> i * 8) := ASCII;	
<b>end loop;</b>	
<b>return</b> slvASCII;	
<b>end function;</b>	
	57

	JS UCiSW 1
<b>Atrybuty obiektów</b>	
E'SIMPLE_NAME is a string containing the name of entity E.	
E'INSTANCE_NAME is a string containing the design hierarchy including E.	
E'PATH_NAME is a string containing the design hierarchy of E to design root.	
<b>Atrybuty sygnałów (na potem)</b>	
S'EVENT true if signal S has had an event this simulation cycle.	
S'STABLE signal: true if no event is occurring on signal S.	
S'STABLE(t) signal: true if no even has occurred on signal S for t units of time.	
S'ACTIVE true if signal S is active during current simulation cycle.	
S'QUIET signal: true if S is quiet. (no event this simulation cycle)	
S'QUIET(t) signal: true if S has been quiet for t units of time.	
S'TRANSACTION bit signal, the inverse of previous value each cycle S is active.	
S'LAST_EVENT the time since the last event on signal S.	
S'LAST_ACTIVE the time since signal S was last active.	
S'LAST_VALUE the previous value of signal S.	
S'DELAYED(t) signal: the value of S at time now - t.	
S'DRIVING true if the process is driving S.	
S'DRIVING_VALUE is the current driving value of signal S in the process.	
	58

	JS UCiSW 1
<b>Atrybuty definiowane przez użytkownika</b>	
-- Declaration:	
<b>attribute</b> Name : AttributeType;	
-- Application:	
<b>attribute</b> Name of ObjectName : ObjectClass is Value;	
ObjectClass = signal   type   function   architecture   ...	
Np.	
Rozpoznawany przez XST atrybut ustalający kodowanie dowolnego typu wyliczeniowego:	
<b>type</b> statetype <b>is</b> (ST0, ST1, ST2, ST3);	
<b>attribute</b> enum_encoding : STRING;	
<b>attribute</b> enum_encoding of statetype : <b>type</b> <b>is</b> "001 010 100 111";	
	59

	JS UCiSW 1
<b>Cykle symulacji</b>	
<b>entity</b> Ex1 <b>is</b>	<b>architecture</b> DFlow of Ex1 <b>is</b>
<b>port</b> (	<b>signal</b> S : STD_LOGIC;
A, B, C : <b>in</b> STD_LOGIC;	<b>begin</b>
Y : <b>out</b> STD_LOGIC );	S <= A <b>or</b> B;
<b>end entity;</b>	Y <= C <b>xor</b> S;
	<b>end architecture;</b>
<b>Testbench:</b>	
<b>entity</b> tbw_Ex1 <b>is</b>	
<b>end</b> tbw_Ex1;	
<b>architecture</b> behavior of tbw_Ex1 <b>is</b>	
<b>component</b> Ex1	<b>signal</b> Y : STD_LOGIC;
<b>port</b> (	<b>begin</b>
A : <b>in</b> STD_LOGIC;	
B : <b>in</b> STD_LOGIC;	ut: Ex1 <b>port</b> map (
C : <b>in</b> STD_LOGIC;	A => A,
Y : <b>out</b> STD_LOGIC	B => B,
);	C => C,
<b>end component;</b>	Y => Y
<b>signal</b> A : STD_LOGIC := '0';	);
<b>signal</b> B : STD_LOGIC := '0';	A <= '1' <b>after</b> 10 ns;
<b>signal</b> C : STD_LOGIC := '0';	<b>end;</b>
	60

**Cykle symulacji** JS UCiSW 1

```

(...)
signal A : STD_LOGIC := '0';
signal B : STD_LOGIC := '0';
signal C : STD_LOGIC := '0';
signal Y : STD_LOGIC;
(...)

```

**TBW:**  
A <= '1' after 10 ns;

**UUT:**  
S <= A or B;  
Y <= C xor S;

	A	B	C	S	Y
Init:	0	0	0	U	U
0ns	0	0	0	U	U
0ns + Δ	0	0	0	0	0
10ns	1	0	0	0	0
10ns + Δ	1	0	0	1	0
10ns + 2Δ	1	0	0	1	1

(KONIEC)

W wykonanie przypisań:  
A <= (...): transakcja '1' / 10 ns → POW\_A  
S <= (...): transakcja '0' / 0 ns → POW\_S  
Y <= (...): transakcja 'U' / 0 ns → POW\_Y

(a) S ← 0 (Event), Y ← U (Active)  
(b) Y <= (...): transakcja '0' / 0 ns → POW\_Y

(a) Y ← 0 (Event)  
(b) null

(a) A ← 1 (Event)  
(b) S <= (...): transakcja '1' / 10 ns → POW\_S

(a) S ← 1 (Event)  
(b) Y <= (...): transakcja '1' / 10 ns → POW\_Y

(a) Y ← 1 (Event)  
(b) null

61

**process ( Clk )** JS UCiSW 1

```

begin
if rising_edge( Clk ) then
Q0 <= Din;
Q1 <= Q0;
end if;
end process;

```

Cykl	Clk	Din	Q0	Q1	Opis:
(...)	'0'	'1'	'0'	'0'	
10ns	'1'	'1'	'0'	'0'	(a) Clk ← 1 (b) Clk'Event, wykonanie procesu: trans. '1' / 10ns → POW_Q0; trans. '0' / 10ns → POW_Q1
10ns + Δ	'1'	'1'	'1'	'0'	(a) Q0 ← 1, Q1 ← 0 (b) Q0'event, Q1 tylko active (koniec)

**process( Clk, Din, Q0, Q1 ) ...?**

```

(...)
10ns + Δ

```

(a) Q0 ← 1, Q1 ← 0  
(b) Q0'Event, wykonanie procesu:  
warunek if niespełniony (koniec)

62

**Atrybuty sygnałów** JS UCiSW 1

S'EVENT true if signal S has had an event this simulation cycle.  
S'STABLE signal: true if no event is occurring on signal S.  
S'STABLE(t) signal: true if no even has occurred on signal S for t units of time.  
S'ACTIVE true if signal S is active during current simulation cycle.  
S'QUIET signal: true if S is quiet. (no transaction this simulation cycle)  
S'QUIET(t) signal: true if S has been quiet for t units of time.  
S'TRANSACTION BIT signal, the inverse of previous value each cycle S is active.  
S'LAST\_EVENT the time since the last event on signal S.  
S'LAST\_ACTIVE the time since signal S was last active.  
S'LAST\_VALUE the previous value of signal S.  
S'DELAYED(t) signal: the value of S at time now - t.

t = 0 ⇒ time is delta

Within a process driving S:  
S'DRIVING true if the process is driving S.  
S'DRIVING\_VALUE current driving value of signal S in the process.

63

**Przykłady** JS UCiSW 1

1) Funkcja rising\_edge():

```

function rising_edge (signal s : STD_ULOGIC) return BOOLEAN is
begin
return (s'EVENT and (To_X01(s) = '1') and
(To_X01(s'LAST_VALUE) = '0'));
end;

```

gdzie

```

CONSTANT cvt_to_x01 : logic x01_table :=
-- 'U' 'X' '0' '1' 'Z' 'W' 'L' 'H' '-'
( 'X', 'X', '0', '1', 'X', 'X', '0', '1', 'X' );

```

2) Wykrycie trwającego co najmniej 100 μs stanu zerowego na linii PS2\_Clk:

```

wait until PS2_Clk'DELAYED'LAST_EVENT > 100 us and
PS2_Clk'LAST_VALUE = '0';

```

64

**Problem: zmiana sygnału WE w momencie jego próbkowania** JS UCiSW 1

**TBW:**  
(...)  
Clk <= not Clk after 100 ns;  
IN\_S <= '1';  
D\_AS <= '1' after 300 ns;  
(...)

Cykl	Clk	D_AS	Q_AS
(...)	'0'	'0'	'0'
300ns	'1'	'1'	'0'

(a) Clk ← 1, D\_AS ← 1  
(b) Clk'Event, wykonanie procesów:  
trans. 1/300ns → POW\_Q\_AS,  
trans. 1/300ns → POW\_Q\_S

(...)

⇒ Nie ma możliwości zdefiniowania przypisania D\_AS w momencie „300 + Δ”

65

**Projekt** JS UCiSW 1

Moduł transkodujący otrzymany bajt na dwa znaki ASCII

```

entity FSM_SendByte is
port ( Clk : in STD_LOGIC;
Reset : in STD_LOGIC;
DI : in STD_LOGIC_VECTOR (7 downto 0);
DIRdy : in STD_LOGIC;
TxBusy : in STD_LOGIC;
DO : out STD_LOGIC_VECTOR (7 downto 0);
DORdy : out STD_LOGIC;
Busy : out STD_LOGIC );
end FSM_SendByte;

```

**architecture RTL of FSM\_SendByte is**

```

type state_type is ( sReset, sReady, sWaitH, sSendH,
sWaitL, sSendL );
signal State, NextState : state_type;
signal regDI : STD_LOGIC_VECTOR (7 downto 0);
signal HalfByte : STD_LOGIC_VECTOR (3 downto 0);

```

66

```

(...)
begin
    -- FSM: State register
    process ( Clk )
    begin
        if rising_edge( Clk ) then
            if Reset = '1' then
                State <= sReset;
            else
                State <= NextState;
            end if;
        end if;
    end process;

    -- FSM: Next state decoding
    process ( State, DIRdy, TxBusy )
    begin
        NextState <= State; -- default

        case State is
            when sReset =>
                NextState <= sReady;

            when sReady =>
                if DIRdy = '1' then
                    NextState <= sWaitH;
                end if;

            when sWaitH =>
                if TxBusy = '0' then
                    NextState <= sSendH;
                end if;

            when sSendH =>
                NextState <= sWaitL;

            when sWaitL =>
                if TxBusy = '0' then
                    NextState <= sSendL;
                end if;

            when sSendL =>
                NextState <= sReady;
        end case;
    end process;
end process;

```

67

```

(...)
-- Outputs
DORdy <= '1' when State = sSendH or State = sSendL
        else '0';

Busy <= '1' when State /= sReady
        else '0';

-- Other, i.e.: input register (with CE)...
regDI <= DI when rising_edge( Clk ) and State = sReady;
-- SKRÓT PROCESU!

-- ...halfByte selection...
HalfByte <= regDI( 7 downto 4 ) when State = sSendH or
        State = sWaitL
        else regDI( 3 downto 0 );

```

68

```

(...)
-- ...transcoding X"0" - X"F" to ASCII '0'-'F'
with HalfByte select
    DO <= X"30" when "0000",
        X"31" when "0001",
        X"32" when "0010",
        X"33" when "0011",
        X"34" when "0100",
        X"35" when "0101",
        X"36" when "0110",
        X"37" when "0111",
        X"38" when "1000",
        X"39" when "1001",
        X"41" when "1010",
        X"42" when "1011",
        X"43" when "1100",
        X"44" when "1101",
        X"45" when "1110",
        X"46" when others;
end RTL;

```

- Nie próbować opisać wszystkiego w jednym procesie
- Każdy sygnał przypisywany w osobnej instrukcji współbieżnej, np. 1 proces / 1 sygnał
- Unikać długich opisów sekwencyjnych, w tym wielokrotnego przypisywania sygnału podczas jednego wykonania procesu
- Jeśli sygnał ma pamiętać swój stan ⇒ rising\_edge( Clk ), bo inaczej będą zatraski

69

## Testbench


```

entity Test_vhd is
end Test_vhd;

architecture behavior of Test_vhd is
    -- component Declaration for the Unit Under Test (UUT)
    component FSM_SendByte
    port(
        Clk : in STD_LOGIC;
        Reset : in STD_LOGIC;
        DI : in STD_LOGIC_VECTOR(7 downto 0);
        DIRdy : in STD_LOGIC;
        TxBusy : in STD_LOGIC;
        DO : out STD_LOGIC_VECTOR(7 downto 0);
        DORdy : out STD_LOGIC;
        Busy : out STD_LOGIC
    );
    end component;

    --Inputs
    signal Clk : STD_LOGIC := '0';
    signal Reset : STD_LOGIC := '0';
    signal DIRdy : STD_LOGIC := '0';
    signal TxBusy : STD_LOGIC := '0';
    signal DI : STD_LOGIC_VECTOR(7 downto 0) := (others=>'0');

```



70

```

--Outputs
signal DO : STD_LOGIC_VECTOR(7 downto 0);
signal DORdy : STD_LOGIC;
signal Busy : STD_LOGIC;

-- AUX
constant Tclk : TIME := 1 us / 50; -- MHz

begin
    -- Instantiate the Unit Under Test (UUT)
    uut: FSM_SendByte port map(
        Clk => Clk,
        Reset => Reset,
        DI => DI,
        DIRdy => DIRdy,
        TxBusy => TxBusy,
        DO => DO,
        DORdy => DORdy,
        Busy => Busy
    );

    -- Global clock 50MHz
    Clk <= not Clk after Tclk / 2;

    -- Reset
    Reset <= '1' after 100 ns, '0' after 100 ns + Tclk;

```

71

```

-- Byte source
process
    type typeByteArray is array ( NATURAL range <> )
        of STD_LOGIC_VECTOR( 7 downto 0 );

    constant arrBytes : typeByteArray
        := ( X"10", X"20", X"3A", X"4F" );

begin
    wait for 200 ns;
    for i in arrBytes'RANGE loop
        if Busy = '1' then
            wait until Busy = '0';
        end if;

        wait for 7.1 * Tclk; -- .1 to avoid rising_edge Clk
        DI <= arrBytes( i );
        DIRdy <= '1';
        wait for Tclk;
        DIRdy <= '0';
    end loop;
    wait; -- forever
end process;

```

72

```
-- ASCII sink
process
begin
  loop
    wait until rising_edge( Clk ) and DORdy = '1';
    TxBusy <= '1';
    wait for 11.1 * Tclk; -- e.g. 11.1 * Tclk
    TxBusy <= '0';
  end loop;
end process;
end architecture;
```